
Online resource 4: NONMEM code 
 
$SIZES NO=10000 LIM6=20000 
$PROBLEM RTTE seizure and dropout model for lacosamide and carbamazepine 
$INPUT ID TYPE TIME DV CL F1X AGE SEX WT NSP3M  
 
$DATA   SimulationData.csv 
IGNORE=@ 
 
$SUBR ADVAN=13 TOL=9 
$MODEL COMP=(SEIZURE) COMP=(DROP) 
 
$PK 
"FIRST  
" USE PRDATA, ONLY: MXSTP01  
" MXSTP01=2147483647  
 
;; Parameters of the RTTE seizure model---------------------------------------- 
 
; 1st event 
TVLAM1 = (THETA(1)) 
TVSHP1 = THETA(2) 
 
; 2nd + events 
TVLAM = (THETA(3)) 
TVSHP = THETA(4) 
; Drug effect on 1st Event 
SLLCM_1 = THETA(11) 
SLLEV_1 = 0 
SLCBZ_1 = THETA(12)  
 
; Drug effect on 2nd+ Events 
SLLCM = THETA(13) 
SLLEV = 0 
SLCBZ = THETA(14)  
 
DEL=1E-12 
;;------------------------------------------------------------------------------ 
;; Parameters of dropout model  ----------------------------------------------- 
;;; VARSEX-DEFINITION START 
IF(SEX.EQ.1) VARSEX = 0  ; Most common - male 
IF(SEX.EQ.0) VARSEX = ( 0 + 0.238) 
;;; VARSEX-DEFINITION END 
 
;;; VARTYPE-DEFINITION START 
VARTYPE = 0  ; CBZ  
IF(TYPE.EQ.3) VARTYPE = ( 0 - 0.138) ; LCM 
;;; VARTYPE-DEFINITION END 
 
;;; VAR-RELATION START 
VARDR=VARSEX +  VARTYPE 
;;; VAR-RELATION END 
 
TM1  = 20 
TM2  = TM1 + 9.76 



TM3  = TM2 + 105 
TM4  = TM3 + 340 
 
GAM = 50 
 
LAM1 = EXP(-6.19) 
LAM2 = EXP(-5.63) 
LAM3 = EXP(-6.96) 
LAM4 = EXP(-7.4) 
LAM5 = EXP(-6.11) 
;;----------------------------------------------------------------------------- 
 
;; Initialization of counters etc.--------------------------------------------- 
IF(NEWIND.NE.2) THEN     ; re-initialize for every new ID 
  TP=0 
  TPEVAL=0  
  CDL = 1      ; current dose level (DL) 
  TIMEDL = 0   ; Time at start of DL 
  ESC = 0      ; Escalation indicator  
  SCOUNTDL= 0  ; seizure counter in dose level 
  SCOUNT = 0   ; total seizure counter 
  DRP=0        ; dropout indicator 
  M6FREE=0     ; seizure free for 6 months counter 
  EVSEIZ= 0    ; counts if seizure in evaluation period 
  RTTE=0 
ENDIF 
;; ---------------------------------------------------------------------------- 
 
IF(ESC.EQ.1.AND.CDL.LT.3) THEN 
   CDL=CDL+1 ; Count DL 1 up upon event only during evaluation period 
   TIMEDL = TIME 
   TPEVAL  = 0 ; start of evaluation period 
   SCOUNTDL= 0 
   EVSEIZ=0   
ENDIF 
TIMEDL1 = TIME - TIMEDL 
 
;; Simulation of AUC  
; takes the individual PK parameters and body weight from the input dataset 
 
;;;; For CBZ ;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;; 
IF(TYPE.EQ.2) THEN  
 IF(CDL.EQ.1) THEN ; CBZ 1st dose step 
  CDOSE = 400 
  IF(TIMEDL1.LT.7) CDOSE= 200 ; 1 weeks up-titration 
 ENDIF  
 IF(CDL.EQ.2) THEN ; CBZ 2nd dose step 
  CDOSE = 800 
  IF(TIMEDL1.LT.7) CDOSE= 600 ; 1 weeks up-titration 
 ENDIF  
 IF(CDL.EQ.3) THEN ; CBZ 3rd dose step 
  CDOSE = 1200 
  IF(TIMEDL1.LT.7) CDOSE= 1000 ; 1 weeks up-titration 
 ENDIF 
 CL2=5.35*(((CDOSE/WT)/15)**0.591)*((WT/70)**0.564) 
 CAUC=CDOSE*F1X/CL2 
ENDIF 
;;;;; FOR LCM ;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;; 
IF(TYPE.EQ.3) THEN 
 IF(CDL.EQ.1) THEN ; LCM 1st dose step 
  CDOSE = 200 
  IF(TIMEDL1.LT.7) CDOSE= 100 ; 2 weeks up-titration 
 ENDIF 
 IF(CDL.EQ.2) THEN       ; LCM 2nd dose step 
  CDOSE = 400 
  IF(TIMEDL1.LT.7) CDOSE= 300 ; 2 weeks up-titration 
 ENDIF 
 IF(CDL.EQ.3) THEN ; LCM 3rd dose step 



  CDOSE = 600 
  IF(TIMEDL1.LT.7) CDOSE= 500 ; 2 weeks up-titration 
 ENDIF 
 CAUC=CDOSE*F1X/CL 
ENDIF 
 
;;; AUC-EFFECT 
IF(TYPE.EQ.2) THEN 
  VARAUC=(0 + SLCBZ * (CAUC-132))  ; if CBZ 
  VARAUC_1= (0 +  SLCBZ_1 * (CAUC-132)) 
ENDIF 
IF(TYPE.EQ.3) THEN 
  VARAUC=(0 + SLLCM * (CAUC-104))  ; if LCM 
  VARAUC_1= (0 +  SLLCM_1 * (CAUC-104)) 
ENDIF 
;;; AUC-DEFINITION END 
 
;;; NSP3M-EFFECT on 1st seizure 
IF(NSP3M.LE.1) VARNSP3M1 = 0 + THETA(5)                    ; 1 seizures 
IF(NSP3M.GE.2.AND.NSP3M.LE.6) VARNSP3M1 = ( 0 )            ; 2-6 seizures 
IF(NSP3M.GE.7.AND.NSP3M.LE.50) VARNSP3M1 = ( 0 + THETA(6)) ; 7-50 seizures 
IF(NSP3M.GT.50) VARNSP3M1 = ( 0 + THETA(7))                ; More than 50 seizures 
;;; VARNSP3M-DEFINITION END 
 
;;; NSP3M-EFFECT on 2nd+ seizures 
IF(NSP3M.LE.1) VARNSP3M = 0 + THETA(8)                    ; 1 seizures 
IF(NSP3M.GE.2.AND.NSP3M.LE.6) VARNSP3M = ( 0 )            ; 2-6 seizures 
IF(NSP3M.GE.7.AND.NSP3M.LE.50) VARNSP3M = ( 0 + THETA(9)) ; 7-50 seizures 
IF(NSP3M.GT.50) VARNSP3M = ( 0 + THETA(10))               ; More than 50 seizures 
;;; VARNSP3M-DEFINITION END 
 
;;; AGE-EFFECT 
SLAGE_1 = 0 
IF(TYPE.EQ.3) THEN  
SLAGE_1 = THETA(17) ; age effect only for LCM 
ENDIF 
 
;; AGE-LAM1 relationship 
VARAGE_1= (0 +  SLAGE_1 * (AGE-41)) 
;;; AGE EFFECT END 
 
 
;; Seizure model --------------------------------------------------------------- 
IF(SCOUNT.LT.1) THEN ; For 1st event 
 LAM = EXP(TVLAM1+ VARNSP3M1 + VARAUC_1 + VARLEV1 + VARAGE_1) ; scale factor  
 SHP = TVSHP1 
ELSE     ; 2nd+ events 
 LAM = EXP(TVLAM + ETA(1) + VARNSP3M + VARAUC + VARLEV) ; scale factor  
 SHP = TVSHP 
ENDIF 
 
$DES 
DADT(1)=LAM*SHP*(LAM*(T-TP)+DEL)**(SHP-1)  ; Hazard of seizure 
 
;; Dropout model equations ----------------------------------------------------- 
FLG5 =  (T+DEL)**GAM/(TM4**GAM+(T+DEL)**GAM)              ; T > TM 4 
FLG4 = ((T+DEL)**GAM/(TM3**GAM+(T+DEL)**GAM)    - 
(T+DEL)**GAM/(TM4**GAM+(T+DEL)**GAM)   )  ; T between TM 3 and 4 
FLG3 = ((T+DEL)**GAM/(TM2**GAM+(T+DEL)**GAM)    - 
(T+DEL)**GAM/(TM3**GAM+(T+DEL)**GAM)   )  ; T between TM 2 and 3 
FLG2 = ((T+DEL)**GAM/(TM1**GAM+(T+DEL)**GAM)    - 
(T+DEL)**GAM/(TM2**GAM+(T+DEL)**GAM)   )  ; T between TM 1 and 2 
FLG1 = (1-(T+DEL)**GAM/(TM1**GAM+(T+DEL)**GAM)   )           ; T<TM 1 
 
HAZD = ( LAM1*FLG1 + LAM2*FLG2 + LAM3*FLG3 + LAM4*FLG4 + LAM5*FLG5 ) * EXP(VARDR) 
 
DADT(2)=HAZD  ; Hazard of dropout 
;; ---------------------------------------------------------------------------- 



 
$ERROR 
XCHZ =  A(1)  ; Cumulative hazard from time=0 
 
;----------Seizure probability and Hazard ------------------------------------- 
IF(NEWIND.NE.2) OLDCHZ=0 ;Initialization of OLDCHZ for each ID 
CHZ = XCHZ-OLDCHZ ;Cumulative hazard since latest event 
SUR = EXP(-CHZ)   ;Seizure probability 
 
HAZ=LAM*SHP*(LAM*(TIME-TP)+DEL)**(SHP-1)  
 
 
;--------- Dropout probability -------------------------------------------------  
CHZDR  = A(2) 
SURDR = EXP(-CHZDR) 
;------------------------------------------------------------------------------- 
 
;--------- Simulation of events ----------------------------------------------- 
IF(ICALL.EQ.4) THEN ; for simulation 
 
;;; DROPOUT EVENTS ;;;;; 
  IF(NEWIND.NE.2) THEN ; draw random number for dropout once per subject 
    CALL RANDOM(3,R) ; for dropout 
    RDRP = R 
  ENDIF 
 
  IF(DRP.EQ.1) THEN  ; check previous dropout status 
      RTTE = 0  
      DV= 0 
    ELSE 
    
   IF(RDRP.GT.SURDR) THEN  ; dropout not due to lack of efficacy 
     DV=0 
     DRP=1 
     RTTE = 1  
     SCOUNTDL = SCOUNTDL +1  ; count event per dose level     
     SCOUNT = SCOUNT +1  
    ENDIF 
  ENDIF ; end DRP=1 
;;; END DROPOUT EVENTS ;;;; 
 
 
;;; SEIZURE EVENTS ;;;;   
  IF(DRP.EQ.0) THEN ; if no dropout go on with seizure evaluation  
    CALL RANDOM (2,R)  ; For seizures, draw random number at every time point to 
make repeated events independent 
    DV=0 
    RTTE = 0 
    ESC = 0 
   
    IF(TIME.EQ.385.AND.CDL.EQ.1) THEN  
      RTTE = 1 ; for censor observations after 121 weeks when DL=1 
      SCOUNTDL = SCOUNTDL +1  ; count event per DL 
      SCOUNT = SCOUNT +1  
    ENDIF 
   
    IF(TIME.EQ.588.AND.CDL.EQ.2) THEN  
      RTTE = 1 ; for censor observations after 121 weeks when DL=1 
      SCOUNTDL = SCOUNTDL +1  ; count event per DL 
      SCOUNT = SCOUNT +1  
    ENDIF 
   
    IF(TIME.EQ.791.AND.CDL.EQ.3) THEN  
      RTTE = 1 ; for censor observations after 121 weeks when DL=1 
      SCOUNTDL = SCOUNTDL +1  ; count event per DL 
      SCOUNT = SCOUNT +1  
    ENDIF 
   



    IF(R.GT.SUR) THEN   ; Seizure event 
      DV=1 
      RTTE = 1 
      SCOUNTDL = SCOUNTDL +1  ; count event per DL 
      SCOUNT = SCOUNT +1  
      IF(TIMEDL1.GE.21.AND.TIMEDL1.LT.203) THEN   ; Period of length 182 days see 
protocol p38 and SAP p61 
        ESC = 1 ; escalation indicator, only escalate within the evaluation period 
        EVSEIZ = 1  ; indicator for seizure during evaluation period 
      ENDIF 
    ENDIF  ; end R > SUR 
  ENDIF   ; end DRP=0 
 
  ; Set RTTE =0 for time greater the scheduled observation period so that it is 
filtered out by PSN 
  IF(TIME.GT.385.AND.CDL.EQ.1) THEN  
    RTTE = 0 ; for censor observations after 121 weeks when DL=1 
  ENDIF 
   
  IF(TIME.GT.588.AND.CDL.EQ.2) THEN  
   RTTE = 0 ; for censor observations after 121 weeks when DL=1 
  ENDIF 
   
  IF(TIME.GT.791.AND.CDL.EQ.3) THEN  
    RTTE = 0 ; for censor observations after 121 weeks when DL=1 
  ENDIF 
 
ENDIF  ; end ICALL=4  
 
; some time counters 
TEVAL = TIMEDL1- 21       ; time since start of evaluation period 
TDIF  = TIME-TP           ; time since last event 
TDIFEVAL = TEVAL - TPEVAL ; time since last event in evaluation period 
 
IF(DV.EQ.1) THEN 
 TP=TIME ; time of event 
 TPEVAL=TEVAL 
ENDIF 
 
IF(TDIFEVAL.GE.182.AND.EVSEIZ.EQ.0) M6FREE = 1    ; indicator for 6-months seizure 
freedom 
 
$THETA   
(-7.22)    ; lnLAMBDA1 
(0.493)    ; SHAPE1 
(-4.72)    ; lnLAMBDA 
(0.713)    ; SHAPE 
(-1.12)    ; NSP3M<1_LAM1 
(1.94)     ; NSP3M7-50_LAM1 
(3.3)      ; NSP3M>50_LAM1 
(-1.37)    ; NSP3M<1_LAM 
(1.36)     ; NSP3M7-50_LAM 
(2.53)     ; NSP3M>50_LAM 
(-0.00917) ; SLLCM_1 
(-0.00658) ; SLCBZ_1 
(-0.00751) ; SLLCM 
(-0.0153)  ; SLCBZ 
(0 FIX)    ; LCM_1 
(0 FIX)    ; LCM 
(-0.0256)  ; SLAGE_1 
$OMEGA 4.12; IIV_LAM 
 
$SIMULATION (5988566) (39978 UNIFORM) (39979 UNIFORM) ONLYSIM NOPREDICTION SUB=100 
$TABLE      ID TYPE CDOSE CAUC TIME TIMEDL SCOUNT CDL TDIF TP DV DRP M6FREE NOPRINT 
            ONEHEADER FORMAT=SF12.3 FILE=simtab 

